
Descrierea SPICE a surselor comandate şi comutatoarelor 
 
 I. Descrierea SPICE a surselor comandate 
 La modelarea circuitelor electronice pot fi folosite surse comandate, care sunt 
caracterizate prin una din următoarele ecuaţii: 

V=f(V),   I=f(V),   V=f(I),    I=f(I), 
în care funcţiile pot fi specificate într-o varietate destul de largă de moduri, argumentele 
funcţiilor putând fi, în unele cazuri, multidimensionale. Datorită diverselor posibilităti de 
descriere, sursele comandate acoperă practic întregul domeniu de interes în modelare şi 
macromodelare. 
 Funcţiile se clasifică în două categorii. În prima categorie intră funcţiile definite în 
domeniul timp, care modelează dispozitive liniare sau neliniare fără memorie, instantanee, 
fiind descrise prin intermediul expresiilor matematice sau tabelar. În a doua categorie intră 
funcţiile definite in domeniul frecvenţă, care modelează funcţii de transfer ale circuitelor 
liniare cu memorie (circuite care conţin capacităţi şi inductanţe). Acestea pot fi descrise prin 
intermediul transformatei Laplace, tabelar sau ca funcţii de transfer corespunzătoare filtrelor 
Cebîşev. 
 În PSPICE pot fi definite patru tipuri de surse comandate: 
– sursa de tensiune comandată în tensiune (STCT), notată cu E; 
– sursa de curent comandată în tensiune (SCCT), notată cu G; 
– sursa de tensiune comandată în curent (STCC), notată cu H; 
– sursa de curent comandată în curent (SCCC), notată cu F. 
 
 a) Descrierea surselor comandate în tensiune (E şi G) 
 
 Sursele comandate în tensiune (E şi G) au şapte forme generale de descriere. Primele 
două forme sunt acceptate în toate versiunile programului PSPICE, iar celelalte cinci forme 
sunt acceptate începând cu versiunea PSPICE 5.1 şi fac parte din categoria formelor extinse. 
  
Primele două forme generale de descriere a surselor comandate în tensiune sunt următoarele: 
 
1) [E][G]<nume> <N+> <N-> <NC+> <NC-> <câştig> 
2) [E][G]<nume> <N+> <N-> POLY(<n>) <NC1+> <NC1-> [<NC2+> <NC2-> ... 
+<NCn+> <NCn->] <p0> <p1> <p2> <p3> <p4> <p5> … 
 
 Prima formă de descriere corespunde cazului surselor comandate liniar, în care 
tensiunea (curentul) sursei depinde liniar de tensiunea de comandă. Semnificaţia parametrilor 
este următoarea: 
– E sau G reprezintă prefixul corespunzător surselor controlate în tensiune; 
– <nume> reprezintă numele sursei, format din maxim 8 caractere; 
– <N+> şi <N-> sunt nodurile (+) si (-) din circuit, între care este conectată sursa comandată; 
– <NC+> şi <NC-> reprezintă nodurile (+) şi (-) corespunzătoare tensiunii de comandă; 
– <câştig> reprezintă valoarea amplificării (transferului). 
 
 A doua formă de descriere corespunde surselor comandate neliniar, polinomial. În 
acest caz tensiunea (curentul) sursei depinde de una sau mai multe tensiuni de comandă, după 
o expresie polinomială. Parametrii din linia de descriere au următoarea semnificaţie: 

– POLY(<n>) reprezintă specificaţia pentru sursele neliniare, polinomiale, unde <n> 
reprezintă numărul de dimensiuni ale polinomului, adică numărul tensiunilor de comandă; 
– <NC1+> <NC1-> [<NC2+> <NC2-> … <NCn+> <NCn->] reprezintă cele n perechi de 
noduri corespunzătoare celor n tensiuni de comandă; 
– <p0> <p1> <p2> <p3> … reprezintă coeficienţii expresiei polinomiale. Expresiile 
polinomiale sunt disponibile pentru orice grad şi orice dimensiune. 
 
 Observaţie: Expresiile polinomiale pot fi implementate mult mai comod, utilizând 
specificaţia VALUE, care va fi prezentată în paragraful următor. 
 
 Considerând nV,...,V,V 21  tensiunile de comandă şi outV  tensiunea comandată, expresia 
polinomială pentru outV  în cazul general este: 

...VpVp...VVp...VVpVp

VVp...VVpVpVp...Vpp)V,...,V,V(V

n)!n(
!nn

n)!n(
!nnnnn

nnnnnnnnout

+⋅+⋅++++++

++++++++=

++−+−
+++

+++

3
1

1222

2

222
2133222

2
212

1212
2

1111021

 

 Particularizând pentru cazul unidimensional (o tensiune de comandă), bimensional 
(două tensiuni de comandă), respectiv tridimensional (trei tensiuni de comandă), expresiile 
polinomiale sunt: 

Exemple: 
*Surse de curent controlate in tensiune 
GSENSE 1 2 10 20 10 ;            GSENSE=10*(V(10)-V(20)) 
Gamp 13 0 POLY(1) 12 0 0 500 ;   Gamp=500*V(12) 
Gneliniar 100 101 POLY(2) 1 2 3 4 0 13.6 0.2 0.005 
*Gneliniar=13.6*(V(1)-V(2))+0.2*(V(3)-V(4))+0.005*(V(1)-V(2))*(V(3)-V(4)) 
*Surse de tensiune controlate in tensiune 
E1 1 2 10 20 10 ; E1=10*(V(10)-V(20)) 
EAMP 13 0 POLY(1) 7 0 0 500 ;    EAMP=500*V(7) 
Enel 100 101 POLY(2) 1 2 3 4 0 13.6 0.2 0.005 
*Enel=13.6*(V(1,2))+0.2*(V(3,4))+0.005*(V(1,2))*(V(3,4)) 
 
 Formele de descriere extinse pentru sursele comandate în tensiune E şi G apar 
începând cu varianta PSPICE 5.1 şi au sintaxa: 
           
E<nume> <N+> <N-> <cuvânt_cheie_ABM>=<funcţia_ABM> 
G<nume> <N+> <N-> <cuvânt_cheie_ABM>=<funcţia_ABM> 
 
unde: 
– <nume> este numele sursei format din maxim 8 caractere; 
– <N+> şi <N-> sunt nodurile din circuit (+) si (-) între care este conectată sursa comandată; 
– <cuvânt_cheie_ABM> (Analog Behavioral Modeling) specifică tipul funcţiei de transfer şi 
poate fi: 
 VALUE - expresie aritmetică; 
 TABLE - tabel puncte de frângere pentru caracteristici liniare pe porţiuni; 
 LAPLACE - transformată Laplace. 

 – <funcţia_ABM> specifică funcţia de transfer în formatul cerut de  <cuvânt_cheie_ABM>  
(formulă matematică sau tabel). 
 
 Sursa de tip E (STCT) are ieşirea în tensiune, iar sursa de tip G (SCCT) are ieşirea în 
curent. Descrierea funcţiei de transfer poate conţine, doar în aceste cazuri, orice combinaţie 
de tensiuni şi curenţi ca argumente. Pentru aceste tipuri de surse nu mai există o diferenţiere 
netă după tipului semnalului de comandă, iar sursele E şi G pot fi, aşadar, comandate atât de 
tensiuni cât şi de curenţi, în acelaşi timp. 
 
 Observaţie: Sursele de tip F şi H nu admit forme extinse de descriere, însă acest lucru 
nu reprezint o limitare. 
 
 Pentru sursele comandate descrise prin intermediul cuvintelor cheie VALUE şi 
TABLE, valoarea ieşirii sursei este furnizată instantaneu, la fiecare moment de timp, această 
descriere fiind specifică funcţiilor de transfer ale circuitelor neliniare fără memorie. 
 
 Observaţie: Pentru analiza în domeniul frecvenţă (.AC), un dispozitiv neliniar este 
liniarizat în punctul static de funcţionare şi este folosit modelul echivalent de semnal mic. 
 
 Cele 3 forme extinse pentru sursele comandate tip E şi G sunt prezentate pe larg în 
continuare. 
 
Descriere cu expresii matematice (VALUE) 
 
 Sursele comandate, definite cu ajutorul opţiunii VALUE, permit ca funcţia de transfer 
instantanee să fie introdusă sub forma unei expresii matematice. 
 
3) [E][G]<nume> <N+> <N-> VALUE={<expresie>} 
 
Exemple: 
ESORT 5 0 VALUE={5*SQRT(V(3,2))} ;     ESQRT=5*(V(3,2)) 2/1  
GPSK 11 6 VALUE={15M*SIN(6.28*10K*TIME+V(3))} 
*GPSK=15E-3*SIN(6.28E4*TIME+V(3)) 
 
 În <expresie> pot fi introduse constante, parametri, tensiuni, curenţi şi timp (TIME). 
Tensiunile pot fi tensiuni de noduri, de exemplu V(3), sau diferenţe de potenţial dintre două 
noduri, de exemplu V(2,3). Curenţii de comandă trebuie să fie curenţi prin surse 
independente de tensiune, de exemplu I(VCC). 
 În descrierea matematică a funcţiilor de transfer (expresiilor) se pot folosi operatorii: 
“+”, “-”, “*”, “/” şi paranteze: “(”, “)”. De asemenea, se pot folosi şi funcţiile matematice 
predefinite, cum ar fi ABS, SQRT, LOG, EXP, SIN, COS, TAN etc., precum şi funcţii 
definite de utilizator cu ajutorul comenzii .FUNC.  
 
Descrierea tabelară prin puncte a caracteristicilor liniare pe porţiuni (TABLE) 
 
 Descrierea tabelară a funcţiei de transfer instantanee permite introducerea datelor 
ridicate experimental sau a modelării liniare pe porţiuni a funcţiilor neliniare. Forma generală 
de descriere a unei surse comandate, care utilizează cuvântul cheie TABLE este: 

4) [E][G]<nume> <N+> <N-> TABLE {<expresie>}=<(<valoare_in>,<valoare_ieş>)>* 
 
unde <expresie> reprezintă, în cazul general, o expresie matematică cu aceeaşi formă ca în 
cazul surselor definite cu ajutorul cuvântul cheie VALUE. În marea majoritatea aplicaţiilor, 
<expresie> se utilizează sub forma cea mai simplă şi anume se reduce doar la tensiunea de 
comandă {V(<NC+>,<NC->)}. 
<(<valoare_in>,<valoare_ieş>)>* reprezintă coordonatele punctelor de frângere ale 
caracteristicii de transfer. 
 
Exemplu: 
*Functia de transfer TABLE 
GRES 1 0 TABLE {V(1)}=(0,0)(0.1,3m)(0.25,3.5m)(0.3,2m)(0.4,2.2m)(0.6,5m) 
Vin 1 0 0.1 
.DC Vin 0 0.6 10m 
.PROBE 
.END 
 
 În figura alăturată este prezentată 
caracteristica statică I-V a unei diode tunel, 
care este modelată liniar pe porţiuni de sursa 
comandată GTUNEL din exemplul 
precedent. Sursa GTUNEL este comandată 
de tensiunea de la bornele sale. Exemplul 
evidenţiază facilitatea de modelare liniară pe 
porţiuni, pe baza datelor experimentale. 
 
 
 Din punct de vedere matematic, pe baza punctelor definite în tabel, se face o 
interpolare liniară cu ajutorul căreia, după evaluarea <expresie>, se calculează valoarea de 
ieşire. Cuvântul TABLE trebuie urmat de blanc (spaţiu). Pentru valorile <expresie> în afara 
domeniului definit în tabel, valoarea ieşirii este constantă şi egală cu prima ieşire din tabel 
când intrarea este mai mică decât prima intrare din tabel, respectiv este constantă şi egală cu 
ultima ieşire din tabel când intrarea este mai mare decât ultima intrare din tabel. 
 
 Observaţie: TABLE poate fi privită ca o extensie pentru VALUE dacă se ţine cont că 
intrarea poate fi o expresie matematică. 
 
 În afară de implementarea funcţiilor obţinute experimental, prin puncte, TABLE este 
împreună cu VALUE deosebit de utilă la implementarea unor modele instantanee ale unor 
dispozitive neliniare, care nu sunt implementate standard. 
 
Funcţii de transfer liniare, cu memorie (modelarea în domeniul frecvenţă) 
 
 Relaţiile care definesc astfel de dispozitive nu sunt instantanee, valoarea de ieşire a 
unei astfel de funcţii de transfer depinzând atât de valorile instantanee ale intrărilor, cât şi de 
memoria acestora. În domeniul timp, această comportare se traduce matematic prin 
modelarea sistemului cu un set de ecuaţii diferenţiale liniare, ceea ce, prin intermediul unei 
transformări integrală de tip Laplace sau Fourier, conduce în domeniul frecvenţă la o expresie 
algebrică. Din acest motiv se preferă pentru această clasă de funcţii de transfer definirea lor în 
domeniul frecvenţă. 

 Observaţie: În timpul analizei .AC se evaluează valoarea ieşirii la fiecare frecvenţă de 
lucru. În timpul analizei .DC se evaluează valoarea ieşirii la frecvenţa 0 Hz (curent continuu). 
În timpul analizei .TRAN se evaluează răspunsul la impuls în domeniul timp pentru funcţia 
de transfer definită în frecvenţă, după care se calculează valoarea ieşirii ca produs de 
convoluţie în domeniul timp dintre intrare şi răspunsul la impuls al funcţiei de transfer. 
Trecerea din domeniul timp în domeniul frecvenţă şi invers poate cauza erori serioase dacă 
nu se face cu mare atenţie. 
 
Transformata Laplace (LAPLACE) 
 
 Descrierea prin intermediul transformatei Laplace a funcţiei de transfer implică 
folosirea în cadrul expresiei a variabilei complexe laplaciene “s”. Cuvântul cheie LAPLACE 
indică faptul că sursa comandată are o funcţie de transfer descrisă prin intermediul 
transformatei Laplace. Intrarea pentru funcţia de transfer este dată de <expresie>, construită 
identic ca în cazul VALUE. Forma de descriere a surselor comandate, ce utilizează cuvântul 
cheie LAPLACE este următoarea: 
 
5) [E][G]<nume> <N+> <N-> LAPLACE {<expresie>}={<transformata>} 
 
Exemplu: 
ERC 5 0 LAPLACE {V(10)}={1/(1+0.001*s)} 
*ERC=1/(1+0.001*s)*V(10) 
 
 Observaţie: În cadrul transformatei Laplace nu pot apărea tensiuni, curenţi sau TIME. 
<transformata> reprezintă expresia în variabila “s” a funcţiei de transfer, aşa cum se poate 
observa în exemplu. 
 Pentru analiza .AC, valoarea <expresie> va fi liniarizată în punctul static de 
funcţionare. <expresie> din linia de descriere a sursei reprezintă intrarea pentru funcţia de 
transfer. În cazul cel mai simplu şi cel mai utilizat de altfel, <expresie> se poate reduce la o 
singură tensiune, care va fi tensiunea de comandă a sursei (de exemplu V(3,8)). Apariţia 
<expresie> ca intrare pentru funcţia de transfer poate, într-un anumit context, să permită o 
exprimare condensată a unor modele de dispozitive neliniare cu memorie. Modelarea acestor 
dispozitive se face prin cascadarea unei funcţii de transfer neliniare, fără memorie 
(echivalentă VALUE), cu o funcţie de transfer liniară, cu memorie (pur LAPLACE). 
 
Exemplu: 
ELAPL_NL 1 0 LAPLACE {V(10)*I(VIN)}={1/(1+0.001*s)} 
*ELAPL_NL=1/(1+0.001*s)*(V(10)*I(VIN)) 
 
 b) Descrierea surselor comandate în curent (H şi F) 
 
 Pentru sursele comandate în curent există numai două forme generale de descriere, 
aceste forme fiind similare cu primele două forme de descriere a surselor comandate în 
tensiune: 
 
1) [F][H]<nume> <N+> <N-> <nume_V_control> <câştig> 
2) [F][H]<nume> <N+> <N-> POLY(<n>) <nume_V_control>* <p0> <p1> <p2> … 
 
unde: 

– F şi H reprezintă prefixul care specifică tipul sursei (H pentru STCC şi F pentru  SCCC); 
– <nume> este numele sursei format din maxim 8 caractere; 
– <N+> şi <N-> sunt nodurile (+) si (-) între care este conectată sursa comandată; 
– <nume_V_control> este numele sursei independente de tensiune (posibil cu valoare nulă), 
prin care circulă curentul de comandă; 
– <câştig> este valoarea amplificării (transferului) şi corespunde cazului surselor liniare; 
– POLY(<n>) reprezintă specificaţia pentru sursele neliniare, polinomiale, unde n reprezintă 
numărul de dimensiuni ale polinomului (numărul de curentenţi de comandă); 
– <nume_V_control>* reprezintă în acest caz numele celor n surse independente de tensiune, 
prin care circulă cei n curenţi de comandă; 
– <p0> <p1> <p2> <p3> … reprezintă valorile coeficienţilor expresiei polinomiale. 
Expresiile polinomiale sunt aceleaşi cu cele prezentate la sursele comandate în tensiune. 
 
Exemple: 
*SURSE DE CURENT CONTROLATE IN CURENT 
Fs 1 2 VSENSE 10 ; Fs=10*I(VSENSE) 
FAMP 13 0 POLY(1) VIN 0 500 ; FAMP=500*I(VIN) 
Fnel 100 101 POLY(2) V1 V2 0 13.6 0.2 0.005 
*Fnel=13.6*I(V1)+0.2* I(V2)+0.005* I(V1)* I(V2) 
* SURSE DE TENSIUNE CONTROLATE IN CURENT 
H1 1 2 VSENSE 10 ; H1=10*I(VSENSE) 
HAMP 13 0 POLY(1) VIN 0 500 ; HAMP=500*I(VIN) 
Hnel 100 101 POLY(2) Vcon1 Vcon2 0 13.6 0.2 0.005 
*Hnel=13.6*I(Vcon1)+0.2* I(Vcon2)+0.005* I(Vcon1)* I(Vcon2) 
 
 Observaţii: Dacă pe latura prin care circulă curentul de comandă nu se află nici o 
sursă independentă de tensiune, atunci pe latura respectivă se va introduce o sursă 
independentă de tensiune de valoare nulă (V=0). 
 
 Sursele comandate în curent nu au forme extinse de descriere, dar acest lucru nu 
constituie oproblemă deoarece se pot folosi formele extinse de descriere a surselor E şi G.

 II. Descrierea SPICE a comutatoarelor  
 
  Un comutator ideal are în starea ON (închis) o rezistenţă nulă, iar în starea OFF 
(deschis) o rezistenţă infinită. Această comportare poate fi aproximată satisfăcător prin 
rezistenţe RON şi ROFF, care sunt semnificativ mai mici, respectiv mai mari decât toate 
celelalte rezistenţe din circuit. Ca urmare, comutatorul este un tip special de rezistor controlat 
în tensiune sau în curent, a cărui rezistenţă variază continuu între RON şi ROFF. 
 
 
 
 
 
 
 
 
 
 
 
 
 
 Forma generală de descriere a comutatorului comandat în tensiune (S), respectiv în 
curent (W) este următoarea: 
 
S<nume> <N+> <N-> <NC+> <NC-> <nume_model> 
W<nume> <N+> <N-> V<nume> <nume_model> 
 
unde:  
– S, W identifică un comutator comandat în tensiune, respectiv în curent; 
– <N+>,< N-> indică nodurile (+) şi (-) între care este contectat comutatorul; 
– <NC+>, <NC-> indică nodurile (+) şi (-) între care se află conectată tensiunea de comandă 
(în cazul comutatorului controlat în tensiune); 
– V<nume> este numele sursei de tensiune prin care circulă curentul de comanda (în cazul 
comutatorului controlat în curent); 
– <nume_model>  este numele modelului care conţine parametrii comutatorului. 
 
 Modelul unui comutator controlat în tensiune, respectiv în curent se defineşte cu 
ajutorul comenzii .MODEL astfel: 
 
.MODEL <nume_model> VSWITCH [parametri_model] 
.MODEL <nume_model> ISWITCH [parametri_model] 
 
Parametrii de model ai comutatoarelor sunt: 
 
Parametru 
de model 

Descriere Unitate 
măsură 

Valoare 
predefinită 

Tip model 

VON Tensiunea de control pentru starea ON Volt 1 VSWITCH 
VOFF Tensiunea de control pentru starea OFF Volt 0 VSWITCH 
ION Curentul de control pentru starea ON Amper 1E-3 ISWITCH 

IOFF Curentul de control pentru starea OFF Amper 0 ISWITCH 
RON Rezistenţa ON Ohm 1.0 VSWITCH,

ISWITCH 
ROFF Rezistenţa OFF Ohm 1E+6 VSWITCH,

ISWITCH 
 
În figura de mai jos este prezentată dependenţa rezistenţei comutatorului controlat în tensiune 
în funcţie de  tensiunea de comandă: 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
Exemple: 
 
S1 3 4 1 2 SMOD 
W2 10 11 V1 WMOD 
.MODEL SMOD VSWITCH (RON=0.1 ROFF=100MEG VON=0.1 VOFF=0) 
.MODEL WMOD ISWITCH (RON=0.1 ROFF=100MEG ION=1mA IOFF=1uA) 

Aplicaţia 1 
 
 Să se scrie fişierul de intrare PSPICE pentru circuitul cu surse comandate de mai jos. 
 
– Sursele independente V1, V2, V3 furnizează formele de undă prezentate în Figura 2; 
– Sursa E4 se va descrie utilizând forma de descriere polinomială (POLY), iar sursa E5 se va 
descrie folosind forma de descriere cu VALUE; 
– Tensiunea furnizată de sursa E7 depinde de sursa V1 după caracteristica din Figura 1; 
– Circuitul va fi analizat în domeniul timp folosind următoarea comandă: 
 .TRAN 30u 2m 0 10u 
 

 
 
 
 
 
 
 
 
 
 
 

Figura 1 – Circuitul cu surse comandate şi caracteristica de transfer a sursei E7. 

Figura 2 – Formele de undă a tensiunilor surselor independente 
 
 În urma descrierii circuitului, efectuaţi simularea şi vizualizaţi formele de undă a 
tensiunilor furnizate de sursele independente precum şi cele furnizate de sursele comandate. 
 
Aplicaţia 2 
 
 Să se scrie fişierul  de intrare PSPICE pentru circuitul din Figura 3, în care sursa Erc 
este comandată de sursa Vin după expresia: inV*H(s)=ErcV , unde H(s) este funcţia de 
transfer a filtrului RC: )s/()s(H α+α= , )C*R/( 111=α . 
 Sursei Vin  i se va specifica numai valoarea de semnal mic (AC 1). Circuitul va fi 
analizat în domeniul frecvenţă folosind următoarea comandă: 
 .AC DEC 10 1 10MEG 
 În urma analizei vizualizaţi caracteristicile de transfer în frecvenţă (diagramele Bodé) 
ale filtrului şi ale sursei Erc: VDB(2), VDB(3), VP(2), VP(3). Verificaţi dacă aceste 
caracteristici se suprapun. 
 

Figura 3 – Circuit RC şi sursă comandată definită în frecvenţă 


